Model Pembelajaran Grup Siklis dan Dihendral dengan Maple*

oleh: Sahid
Jurdik Matematika FMIPA UNY

Abstrak

Diantara topik-topik yang diajarkan dalam matakuliah Aljabar Abstrak adalah grup siklis dan grup dihedral. Kedua topik ini biasanya didefinisikan dan disajikan secara abstrak dalam banyak buku teks, sehingga terdapat kemungkinan mahasiswa mengalami kesulitan memahami dan menginterpretasikan konsep-konsep yang memiliki banyak penerapan tersebut dan untuk melakukan eksplorasi sifat-sifat kedua grup.

Kesulitan tersebut dapat diatasi dengan menggunakan software Maple, sebuah paket matematika canggih dan lanjut yang dapat dipakai sebagai alat untuk belajar dan mengajarkan matematika secara mudah. Dengan menggunakan perintah-perintah dan prosedur Maple yang tersedia dan prosedur serta fungsi yang mudah ditulis, para dosen dan mahasiswa dapat melakukan eksplorasi dan penyelidikan sifat-sifat grup siklis dan dihedral.

Artikel ini menunjukkan sebuah model belajar dan mengajarkan kedua topik dengan menggunakan Maple. Model pembelajaran dimulai dari diskripsi abstrak kedua konsep dan dilanjutkan dengan penjelasan bagaimana mendefinisikan dan menggunakan prosedur dan fungsi Maple untuk melakukan komputasi dan visualisasi kedua jenis grup disertai dengan contoh-contoh. Beberapa soal latihan yang diberikan pada akhir proses pembelajaran dapat dipakai untuk mengevaluasi kemajuan belajar mahasiswa.

Kata Kunci: *pembelajaran, grup siklis, grup dihedral, Maple*

Instructional Model of Cyclic and Dihedral Groups Using Maple

Abstract

Among other topics taught in the subject of Abstract Algebra are **cyclic group** and **dihedral group**. These topics are usually defined and presented abstractly in many textbooks, so there is possibility for students to have difficulty to understand and interpret these applicable concepts and to explore their properties.

This difficulty can be handled by using the software Maple, a sophisticated and advanced mathematics package that can be used as a tool for learning and teaching mathematics easily. Using the available Maple commands and procedures and easily defined procedures and functions, teachers and students can explore and investigate the properties of cyclic and dihedral groups.

This article demonstrates a learning and teaching model of those topics using Maple. The model starts with the abstract description about those concepts and proceeds with the description on how to define and to apply the Maple procedures and function to do calculations and visualization of those groups accompanied with some examples. Some exercises given at the end of the instructional process can be used to evaluate the progress of students learning.

Pendahuluan

Artikel ini menyajikan sebuah model pembelajaran grup siklis dan grup dihedral dengan menggunakan Maple. Grup siklis dan grup dihedral merupakan dua buah topik atau contoh grup yang sering dibahas dalam teori grup. Kedua grup tersebut biasanya disajikan secara abstrak dan

---

* Disampaikan pada Seminar Nasional Pendidikan Matematika dan Ilmu Pengetahuan Alam, Universitas Negeri Malang, 5 Agustus 2002

eksplorasi sifat-sifatnya jika dikerjakan secara manual hanya dapat dilakukan secara terbatas, untuk grup-grup berukuran kecil. Untuk mengatasi keterbatasan eksplorasi manual, dapat digunakan alat bantu komputer. Program Maple merupakan salah satu aplikasi matematika yang sangat cocok untuk mengajarkan konsep grup siklis dan grup dihedral. Hal ini dikarenakan Maple menyediakan fasilitas untuk komputasi matematis dan kemampuan visualisasi yang sangat mudah dipakai, baik oleh pengguna pemula maupun pengguna yang sudah mahir.

Secara matematis, konsep **grup siklis** didefinisikan sebagai (Herstein, 1975: 30):

*Misalkan **a** sebarang simbol dan himpunan $\mathbf{C_n}$ terdiri atas semua simbol $\mathbf{a^i}$, **i=0, 1, 2, …, (n-1)**, dengan aturan $\mathbf{a^0 = e}$, $\mathbf{a^i \cdot a^j = a^{i+j}}$ **jika i+j ≤ n** dan , $\mathbf{a^i \cdot a^j = a^{i+j-n}}$ **jika i+j >n**. $\mathbf{C_n}$ dengan operasi perkalian tersebut dinamakan **grup siklis** berorder **n** yang dihasilkan oleh **a** dengan elemen idenitas **e**.*

Konsep **grup dihedral** dapat didefinisikan sebagai (Herstein, 1975: 54):

*Misalkan **r** dan **R** adalah dua simbol yang berbeda dan himpunan $\mathbf{D_n}$ terdiri atas semua simbol $\mathbf{r^iR^j}$, **i, j = 0, 1, 2, …, (n-1)** dengan aturan (1) $\mathbf{r^i.R^j = r^k R^l}$ jika dan hanya jika **i**=k dan **j=l**, (2) $\mathbf{r^2 = R^n = e}$, **n>2**, dan (3) $\mathbf{r.R = R^{n-1}}$ r. $\mathbf{D_n}$ dengan operasi perkalian tersebut dinamakan **grup dihedral** berorder **2n** yang dihasilkan oleh **r** dan **R** dengan elemen idenitas **e**.*

Grup siklis merupakan salah satu grup istimewa dan mempunyai peranan penting dalam teori grup (Herstein, 1975:39).

Permasalahannya adalah bagaimana mengajarkan konsep-konsep grup tersebut, yang bersifat abstrak, kepada mahasiswa agar peserta didik memahaminya dan dapat memberikan interpretasi kedua grup tersebut beserta sifat-sifat dan aplikasinya. Diperlukan penjelasan konsep kedua grup tersebut dengan cara menyajikan realisasi keduanya, sehingga konsep abstrak tersebut dapat terserap oleh peserta didik melalui contoh-contoh nyata dalam bentuk visualisasi. Selanjutnya, visualisasi dapat dilakukan secara mudah dengan menggunakan program Maple.

Mendefinisikan Grup Siklis dan Dihedral

Realisasi Grup Siklis dan Dihedral

Langkah pertama di dalam menyajikan grup siklis dan grup dihedral dengan Maple adalah mencari realisasi kedua grup tersebut. Grup siklis dan dihedral dapat disajikan sebagai himpunan **rotasi** dan **refleksi** (pencerminan) segi banyak beraturan. (Selanjutnya, yang dimaksud segi-n adalah segi-n beraturan.) Untuk mudahnya, setiap titik sudut segi-n diberi nomor secara ber-lawanan arah jarum jam dari 1 sampai n. Rotasi yang dipakai adalah perputaran dengan pusat titik pusat segi-n berlawanan arah dengan arah jarum jam sejauh kelipatan ($360^o / n$) . Refleksi yang dipakai adalah pencerminan terhadap sumbu-sumbu simetri segi-n yang bersangkutan.

Dengan menandai setiap titik sudut segi-n dengan angka 1, 2, …, n, maka baik rotasi maupun refleksi dapat dipandang sebagai transformasi yang memindahkan label titik-titik sudut tersebut ke titik-titik sudut yang lain. Misalnya, dalam segi-5, jika R_2 menyatakan rotasi sebesar

72^o, maka R_2 memindahkan 1 ke 2, 2 ke 3, 3 ke 4, 4 ke 5, dan 5 ke 1. Jika r_7 menyatakan pencerminan terhadap sumbu simetri yang melalui titik 5 dan titik tengah rusuk penghubung 2 dan 3, maka r_7 memindahkan 1 ke 4, 2 ke 3, 3 ke 2, 4 ke 1, dan 5 ke 5. (Lihat Gambar 1).



Gambar 1 Contoh Rotasi dan Refleksi pada Segi-5 Beraturan

Dengan realisasi rotasi dan refleksi (simetri) segi-n tersebut, grup siklis C_n menpunyai elemen rotasi-rotasi sejauh $k(360^o/n)$, untuk k=0, 1, 2, …, (n-1), sedangkan grup dihedral D_n terdiri atas n rotasi dalam C_n dan n refleksi pada sumbu simetri yang melalui titik sudut k, untuk k= 1, 2, 3, …, n. Operasi biner yang dipakai pada kedua grup tersebut adalah "dilanjutkan".

Prosedur Maple untuk Menghasilkan Grup Siklis dan Dihedral

Untuk menyajikan grup siklis dan grup dihedral pada Maple, setiap rotasi dan refleksi dinyatakan dalam bentuk permutasi. Suatu permutasi n obyek dapat dipandang sebagai suatu pemetaan 1-1 yang mengawankan setiap obyek dengan tepat satu obyek lain. Jika terdapat n obyek yang dinya-takan dengan angka 1, 2, 3, .., n, maka setiap permutasi dapat ditulis sebagai [$a_1, a_2, a_3, \dots, a_n$], yang berarti pemetaan 1 ke a_1, 2 ke a_2, 3 ke a_3, …, n ke a_n dengan a_k di dalam {1, 2, 3, ..., n}, $k = 1, 2, 3, \dots, n$; $a_i \neq a_j$, jika $i \neq j$. Sebagai contoh, rotasi ($360^o/n$) pada segi-n yang memindahkans titik sudut 1 ke 2, titik sudut 2 ke 3, dan seterusnya, ..., titik sudut n ke 1, dapat disajikan sebagai permutasi [2, 3, ..., n, 1]. Jika sebuah simetri memindahkan titik sudut 1 ke titik sudut a_1, titik sudut 2 ke titik sudut a_2, titik sudut 3 ke titik sudut a_3, …, titik sudut n ke titik sudut a_n, maka simetri tersebut dapat disajikan sebagai permutasi [$a_1, a_2, a_3, \dots, a_n$]

Rotasi segi-*n* dapat dinyatakan dengan permutasi [*i, i+1, ..., n, 1, 2, .., i-1*], *i=1, 2, ..., n*. Refleksi (simetri) segi-*n* dapat dinyatakan dengan permutasi [*n+1-i, n-i,..., 1, n, n-1, .., n-i+2*], *i=1, 2, ..., n*. Dalam Maple permutasi [*k, k+1, k+2,..., n, 1, 2,..., k-1*] dapat dituliskan sebagai [$k..n,$1..k-1]. Identitas (tidak mengubah titik sudut) disajikan sebagai permutasi [1, 2, ..., n], atau [$1..n].

Catatan:
Notasi dolar ($) dalam Maple digunakan untuk menghasilkan suatu barisan ekspresi, seperti contoh berikut:

```
> $ 2..5;  i^2 $ i = 2/3 .. 8/3; a[k] $ k = 1..3;  x$4;
```

$$2, 3, 4, 5 \qquad \frac{4}{9}, \frac{25}{9}, \frac{64}{9} \qquad a_1, a_2, a_3 \qquad x, x, x, x$$

Dari penjelasan tersebut dapat didefinisikan dua buah fungsi Maple sebagai berikut untuk menghasilkan elemen-elemen (dalam bentuk permutasi) grup siklis dan dihedral pada segi- n (lihat Mihailovs, 2001).

```
> siklis:=n->[seq([$i..n,$1..i-1],i=1..n)]:
> dihedral:=n->[op(siklis(n)),seq([n+1-j$j=i..n,n+1-j$j=1..i-1],
                                    i=1..n)]:
```

Catatan:

- *Fungsi Maple **seq** berguna untuk menghasilkan suatu barisan (satu atau dua dimensi) dengan rumus dan barisan indeks yang diberikan.*
- *Fungsi **op** berguna untuk menampilkan elemen-elemen suatu ekspresi, dalam definisi di atas, elemen-elemen grup siklis.*
- *Untuk mengetahui penjelasan perintah Maple, mahasiswa dapat diminta untuk menuliskan **?fungsi**, dengan **fungsi** adalah nama perintah Maple, misalnya **>?seq**.*

Grup siklis didefinisikan untuk semua bilangan bulat *n,* namun grup dihedral didefinisikan untuk bilangan-bilangan bulat *n* yang tidak kurang 3 (minimal grup simetri segitiga beraturan!).

Mahasiswa diminta untuk menuliskan definisi di atas pada layar Maple (pada baris yang bertanda ">"). Definisi diakhiri dengan menekan tombol ENTER, dan selanjutnya kedua fungsi siap digunakan. Kedua fungsi Maple tersebut berguna untuk menampilkan elemen-elemen grup siklis C_n dan D_n, seperti contoh-contoh di bawah ini. Selanjutnya, mahasiswa dapat menampilkan elemen-elemen kedua grup tersebut untuk nilai-nilai n yang lain.

Contoh

Perintah berikut menghasilkan daftar permutasi yang bukan grup.

```
> dihedral(2);
```

$$[[1, 2], [2, 1], [2, 1], [1, 2]]$$

Berikut adalah definisi grup yang benar.

Grup rotasi segitiga beraturan:

```
> siklis(3);
```

$$[[1, 2, 3], [2, 3, 1], [3, 1, 2]]$$

Grup simetri segitiga beraturan:

```
> dihedral(3);
```

$[[1, 2, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1], [2, 1, 3], [1, 3, 2]]$

Grup simetri bujur sangkar:

```
> dihedral(4);
```

$[[1, 2, 3, 4], [2, 3, 4, 1], [3, 4, 1, 2], [4, 1, 2, 3], [4, 3, 2, 1], [3, 2, 1, 4], [2, 1, 4, 3],$
$[1, 4, 3, 2]]$

Di sini mahasiswa harus memahami arti setiap notasi di atas. Pada grup D_4 (yakni, **dihedral(4)**) di atas misalnya, elemen terakhir $[1, 4, 3, 2]$ berarti permutasi yang memetakan titik sudut 1 ke 1, 2 ke 4, 3 ke 3, dan 4 ke 2, sehingga permutasi tersebut menyajikan refleksi terhadap diagonal pengubung 1 dan 3 (Lihat Gambar 3 di bawah).

Notasi grup permutation di atas agak panjang. Untuk membuat notasi yang lebih mudah (ringkas), setiap elemen dapat dinyatakan dengan bilangan-biangan ordinal (urutannya) dalam grup tersebut, sehingga elemen identitasnya selalu 1, dan elemen terakhir D_4, misalnya, adalah 8.

Dengan notasi baru ini, untuk mencari invers suatu elemen, didefinisikan prosedur berikut (lihat Mihailovs, 2001) pada baris perintah Maple:

```
> invep:=proc(a,G) local i,v,b,k; k:=nops(G[a]); b:=[0$k];
     for i to k do b[G[a][i]]:=i od; member(b,G,'v'); v end:
```

<u>Catatan:</u>

nops : *menghitung cacah unsur yang diopersikan dalam suatu ekspresi*
member : *menguji apakah suatu elemen anggota suatu himpunan atau modul*
local : *mendefinisikan variabel lokal (yang hanya dipakai dalam prosedur yang bersangkutan)*

Contoh

Sebagai contoh, invers elemen ke-5 dalam grup siklis segi-12 adalah elemen ke-9:

```
> invep(5,siklis(12));
```

$$9$$

Invers elemen ke-5 dalam D_4 adalah elemen ke-5:

```
> invep(5,dihedral(4));
```

$$5$$

Mahasiswa dapat diminta untuk mencari invers beberapa elemen lain dalam kedua grup tersebut dengan cara yang sama.

Visualisasi Grup Siklis dan Grup Dihedral

Sampai di sini, mahasiswa hanya diperkenalkan dengan simbol-simbol. Agar mereka lebih memahami sifat-sifat kedua grup tersebut, perlu diberikan visualisasi elemen-elemen kedua grup tersebut. Maple memberikan kemudahan untuk melakukan hal ini

Prosedur **Gambar** berikut ini berguna untuk menampilkan grafik *ks* elemen dari grup G, mulai dari elemen ke-*m*, ditampilkan dalam *k* kolom dan *s* baris. Untuk dapat menggambar grafik diperlukan paket Maple **plots** dan **plottools**, yang dapat dipanggil dengan perintah **with**.

```
> with(plots): with(plottools):
> Gambar:=proc(G,m,k,s) local segin,seginasli,seginhasil,n,a,b,i,
              j1,j2, p, panah,l,label,titikasal,titikhasil,nomor;
n:=nops(G[1]); # (a,b) adalah pusat segi-n beraturan
segin := (a,b)->[seq([a+cos(2*Pi*i/n),b+sin(2*Pi*i/n)],i=1..n)]:
seginasli:=seq(seq(segin(10*j1,-4*j2),j1=1..k),j2=1..s);
seginhasil:=seq(seq(segin(10*j1+4.5,-4*j2),j1=1..k),j2=1..s);
p:=polygonplot({seginasli,seginhasil},axes=NONE,
                scaling=CONSTRAINED, color=aquamarine):
panah:=seq(seq(arrow([10*j1+1.6,-4*j2],vector([1.2,0]),.05,.3,.3,
                                       color=blue),j1=1..k),j2=1..s):
label:=(a,b,l)->seq([a+1.4*cos(2*Pi*i/n),b+1.4*sin(2*Pi*i/n),
                     l[i+1]],i=0..n-1):
```

```
titikasal:=textplot([seq(seq(label(10*j1,-4*j2,[$1..n]),j1=1..k),
                          j2=1..s)],color=black):
titikhasil:=textplot([seq(seq(label(10*j1+4.5,-4*j2,G[invep(m-
1+j1+k*(j2-1),G)]),j1=1..k),j2=1..s)],color=red):
nomor:=textplot([seq(seq([10*j1+2.2,-4*j2+.5,m-1+j1+k*(j2-1)],
                          j1=1..k),j2=1..s)],color=blue):
display(p,panah,titikasal,titikhasil,nomor) end:
```

Mahasiswa diminta secara hari-hati menuliskan prosedur **Gambar** di atas untuk menghindari kesalahan. Seteleh selesai mendefinisikan prosedur **Gambar** di atas, mahasiswa dapat menampilkan elemen-elemen grup siklis dan dihedral secara visual sesuai keinginnya.

Contoh

Berikut adalah elemen-elemen grup siklis (rotasi) bujur sangkar (C_4). Setiap anggota ditunjukkan dengan nomor urutnya dan visualisasi transformasinya.

```
> Gambar(siklis(4),1,2,2);
```



Gambar 2. Visualisasi grup rotasi bujur sangkar (C_4)

Berikut adalah gambar anggota-anggota D_4 .

```
> Gambar(dihedral(4),1,2,4);
```



Gambar 3. Visualisasi grup simetri bujur sangkar (D_4)

Berikut adalah daftar visualisasi anggota-anggota D_6 :

```
> Gambar(dihedral(6),1,2,6);
```



Gambar 4. Visualisasi grup simetri heksagon (D_6)

Mahasiswa juga dapat menampilkan sebuah elemen saja, misalnya anggota ke-11 dalam D_{10} adalah:

```
> Gambar(dihedral(10),11,1,1);
```



Gambar 5. Visualisasi sebuah elemen grup simetri segi-10 (D_{10})

Ingat bahwa grup siklis merupakan subgrup grup dihedral, sehingga untuk menampilkan visualisasi sebuah rotasi segi-n dapat digunakan perintah **siklis** maupun dihedral. Merupakan sebuah rotasi atau refleksikah elemen tersebut? Dengan hanya melihat nomor anggota dan tanpa melihat gambar di sebelah kiri dan kanan sudah dapat diketahui apakah suatu elemen merupakan rotasi atau refleksi.

Mahasiswa dapat diminta untuk menyebutkan permutasi yang menyatakan elemen suatu grup dihedral dengan melihat visualisasinya. Pertanyaan ini dapat dijawab dengan mudah mengguna-kan perintah **siklis** atau **dihedral**.

Contoh

Berikut adalah representasi permutasi dari anggota ke-11 dalam D_{10} (Bandingkan dengan Gambar 5).

```
> dihedral(10)[11];
```

$$[10, 9, 8, 7, 6, 5, 4, 3, 2, 1]$$

Permutasi yang merupakan elemen ke-5 dalam C_7 dapat dicari dengan beberapa cara sebagai berikut.

```
> siklis(7)[5]; dihedral(7)[5];
```

$$[5, 6, 7, 1, 2, 3, 4] \qquad [5, 6, 7, 1, 2, 3, 4]$$

```
> Gambar(siklis(7),5,1,1);
```



Gambar 6. Visualisasi elemen ke-5 grup rotasi segi-7 (C_7)

Dari Gambar 7 di atas, terlihat bahwa pemetaannya adalah 1 ke 5, 2 ke 6, 3 ke 7, 4 ke 1, 5 ke 2, 6 ke 3, dan 7 ke 4. Jadi dalam bentuk permutasi rotasi tersebut adalah [5, 6, 7, 1, 2, 3, 4] (sama dengan kedua hasil sebelumnya).

Mahasiswa dapat melakukan eksplorasi elemen-elemen suatu grup siklis dan dihedral dengan cara di atas. Mahasiswa dapat menampilkan visualiasi sebuah elemen, menyatakan per-mutasinya, dan mengeceknya menggunakan Maple. Dengan demikian mahasiswa akan lebih memahami elemen-elemen kedua grup tersebut.

Dari contoh-contoh di atas, mahasiswa harus menyadari bahwa notasi G[k] berarti elemen ke-k pada grup G. Grup siklis C_n memiliki n elemen dan grup dihedral D_n mempunyai $2n$ elemen. Semua anggota grup siklis adalah rotasi, n elemen pertama **dihedral**(n) juga rotasi, dan n elemen terakhirnya adalah refleksi. Fungsi **nops** dapat digunakan untuk menghitung cacah anggota kedua grup tersebut.

Contoh

Berikut berturut-turut perintah untuk menghitung cacah anggota grup-grup C_{12}, C_{15}, C_{500}, D_{100}, D_{500}, D_{1000}

```
> nops(siklis(12)); nops(siklis(15));nops(siklis(500));
```

$$12 \quad 15 \quad 500$$

```
> nops(dihedral(100));nops(dihedral(500));nops(dihedral(1000));
```

$$200 \quad 1000 \quad 2000$$

Tabel Cayley dan Pengujian Grup Abelian

Tabel *Cayley* adalah tabel perkalian yang memuat elemen-elemen suatu grup/himpunan G (berhingga), yakni tabel yang berisi elemen-elemen $a * b$ untuk semua a dan b anggota G. Tabel ini berguna untuk melihat sifat-sifat suatu grup dan juga untuk menguji apakah suatu himpunan dengan operasi biner (secara umum disebut "perkalian") tertentu merupakan suatu grup.

Untuk menyusun tabel *Cayley* dari himpunan G diperlukan definisi perkalian *. Pada Maple kita dapat mendefinisikan prosedur **kalip** di bawah ini, yang dapat digunakan untuk mengalikan dua buah elemen a dan b (a dan b menyatakan urutan anggota) dari himpunan atau grup G. Mahasiswa diminta untuk menuliskan prosedur **kalip** tersebut.

```
> kalip:=proc(a,b,G) local i,v;
member([seq(G[a][G[b][i]],i=1..nops(G[a]))],G,'v');v end:
```

Contoh

Contoh berikut memberikan ilustrasi bagaimana mengalikan dua buah elemen dari suatu grup dengan menggunakan fungsi **kalip** tersebut:

```
> kalip(3,7,siklis(11)); kalip(7,3,siklis(11));
```

$$9 \quad 9$$

```
> kalip(3,7,dihedral(4)); kalip(7,3,dihedral(4));
```

$$5 \quad 5$$

Perlu dicatat di sini, mahasiswa harus tetap diingatkan bahwa angka-angka di atas menyatakan nomor urut keanggotaan elemen dalam grup yang bersangkutan. Mahasiswa sebaiknya diminta untuk menguji kebenaran hasil perkalian tersebut, dengan melihat permutasi-permutasi yang memiliki nomor-nomor urut tersebut. Sebagai contoh, dalam grup D_4 :

```
> dihedral(4)[7]; # Elemen ke-7
```

$$[2, 1, 4, 3]$$

```
> dihedral(4)[3]; dihedral(4)[5];   # Elemen ke-3 dan ke-5
```

$$[3, 4, 1, 2] \quad [4, 3, 2, 1]$$

Elemen ke-7 adalah permutasi yang memetakan 1 ke 2, 2 ke 1, 3 ke 4, dan 4 ke 3;
Elemen ke-3 adalah permutasi yang memetakan 1 ke 3, 2 ke 4, 3 ke 1, dan 4 ke 2;
sehingga hasilkali keduanya adalah permutasi yang memetakan 1 ke 4, 2 ke 3, 3 ke 2, dan 4 ke 1, yakni [4, 3, 2, 1]. Ternyata sama dengan hasil perhitungan di atas! Hal ini akan lebih jelas dengan menuliskan perintah sebagai berikut.

```
> dihedral(4)[kalip(7,3,dihedral(4))];
```

$$[4, 3, 2, 1]$$

Pertanyaan: *Apakah grup siklis dan grup dihedral bersifat komutatif?* Jawabnya ada di bawah ini.

Dengan menggunakan fungsi **kalip** tersebut mahasiswa dapat menyusun tabel *Cayley* untuk grup-grup berorder kecil, misalnya C_3, C_4, C_5, D_3 dan D_4. Cara yang terbaik untuk menyusun tabel *Cayley* sebuah grup adalah dengan mendefinisikan sebuah prosedur, seperti fungsi Maple **cayley** berikut ini. Mintalah mahasiswa untuk menuliskan fungsi tersebut pada baris perintah Maple.

```
> cayley:=G->Matrix(nops(G),(i,j)->kalip(i,j,G)):
```

Setelah mendefinisikan kedua prosedur di atas (**kalip** dan **cayley**), mahasiswa dapat menampilkan tabel perkalian grup-grup siklis dan dihedral yang dikehendaki.

Contoh

Misalnya tabel *Cayley* untuk D_4 adalah:

```
> d[4]:=cayley(dihedral(4));
```

$$d_4 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 \\ 2 & 3 & 4 & 1 & 8 & 5 & 6 & 7 \\ 3 & 4 & 1 & 2 & 7 & 8 & 5 & 6 \\ 4 & 1 & 2 & 3 & 6 & 7 & 8 & 5 \\ 5 & 6 & 7 & 8 & 1 & 2 & 3 & 4 \\ 6 & 7 & 8 & 5 & 4 & 1 & 2 & 3 \\ 7 & 8 & 5 & 6 & 3 & 4 & 1 & 2 \\ 8 & 5 & 6 & 7 & 2 & 3 & 4 & 1 \end{bmatrix}$$

Ingat, setiap elemen adalah representasi (nomor urut) permutasi dalam D_4. *Nomor 1-4 adalah rotasi dan nomor 5-8 adalah refleksi.*

Catatan:

Dalam penulisan pada Maple di atas digunakan notasi ***d[4]*** *karena notasi* ***D[4]*** *akan menyebabkan error.*

Terlihat pada tabel *Cayley* dalam contoh di atas bahwa komposisi dua refleksi adalah suatu rotasi! Invers suatu refleksi adalah refleksi itu sendiri! Matriks di atas tidak simetris, karena memang D_4 bukan grup *Abelian* (grup komutatif), yakni tidak berlaku bahwa $a * b = b * a$ untuk semua a dan b dalam grup yang bersangkutan. Tabel Cayley grup Abelian bersifat simetris terhadap diagonal utamanya.

Fungsi Maple berikut berguna untuk menguji apakah suatu grup merupakan grup Abelian:

```
> isAbelian:=G->IsMatrixShape(cayley(G),symmetric):
```

<u>Catatan:</u>
*Fungsi Maple **IsMatrixShape** berguna untuk mengecek apakah suatu matriks simetris atau tidak.*

Contoh
Dari contoh tabel Cayley sebelumnya sudah diketahui bahwa grup D_4 tidak bersifat komutatif. Hal ini dapat juga diuji dengan perintah sebagai berikut.

```
> isAbelian(dihedral(4));
```

$$false$$

Tabel Cayley grup siklis bersifat sangat simmetris:

```
> C[5]:=cayley(siklis(5));
```

$$C_5 := \begin{bmatrix} 1 & 2 & 3 & 4 & 5 \\ 2 & 3 & 4 & 5 & 1 \\ 3 & 4 & 5 & 1 & 2 \\ 4 & 5 & 1 & 2 & 3 \\ 5 & 1 & 2 & 3 & 4 \end{bmatrix}$$

Jadi grup C_5 adalah grup Abelian:

```
> isAbelian(siklis(5));
```

$$true$$

Dengan tanpa melihat tabel Cayley-nya dapat diketahui apakah grup-grup C_{15}, C_{20}, D_{23}, D_{30} dan D_{50} bersifat komutatif atau tidak.

```
> isAbelian(siklis(15)); isAbelian(siklis(20));
```

$$true \quad true$$

```
> isAbelian(dihedral(23)); isAbelian(dihedral(30));
```

$$false \quad false$$

```
> isAbelian(dihedral(50));
```

$$false$$

Pertanyaan: *Dapatkah disimpulkan bahwa grup siklis adalah grup Abelian dan grup dihedral non-Abelian?*

Dengan menggunakan prosedur-prosedur (fungsi) Maple di atas, mahasiswa dapat melakukan eksplorasi dan penyelidikan sifat-sifat grup siklis dan dihedral secara mudah dengan cara mengamati tabel Cayley grup yang bersangkutan dan melakukan perhitungan-perhitungan yang sesuai. Kepada mahasiswa dapat diberikan beberapa tantangan pemakaian prosedur-prosedur di atas, misalnya:

1. mencari invers suatu rotasi, hasilkali rotasi dan refleksi dan inversnya
2. mencari periode suatu elemen grup siklis dan grup dihedral
3. mencari subgrup dari suatu grup siklis atau dihedral
4. mencari subgrup normal dari suatu grup siklis atau dihedral
5. menyelidiki sifat-sifat subgrup dari grup siklis dan dihedral

6. mencari koset sebuah subgrup dari suatu grup siklis atau dihedral, dan lain-lain.

Operasi Perkalian dan Inversnya

Di atas telah dijelaskan bagaimana mengalikan dua buah elemen suatu grup dengan menggunakan fungsi **kalip**, yang sudah didefinisikan sebelumnya pada baris Maple. Cara lain untuk melakukan hal ini adalah dengan mendefinisikan **operator** perkalian dan invers perkalian. Berikut ini disajikan bagaimana mendefinisikan kedua operator tersebut.

Apabila kita perlu melakukan banyak perhitungan (perkalian) dalam suatu grup tertentu, misalnya D_4, kita dapat mendefinisikan perkalian khusus dan operasi invers elemen terhadap perkalian tersebut dengan cara sebagai berikut.

Operator perkalian pada suatu **Grup** dapat didefinisikan dengan cara menuliskan:

```
> `&*`:=(a,b)->kalip(a,b,Grup):
```

Invers elemen suatu **Grup** didefinisikan sebagai berikut:

```
> `&-`:=a->invep(a,Grup):
```

Catatan:
Tanda & digunakan untuk mengawali operator bebas dalam Maple.

Perhatikan, definisi kedua operator di atas menggunakan definisi fungsi/prosedur **kalip** (untuk perkalian) dan **invep** (untuk invers). Untuk menggunakan operasi tersebut, kita harus menentukan grupnya dengan nama "**Grup".**

Contoh
Misalkan kita ingin melakukan perhitungan-perhitungan pada grup D_4.

```
> Grup:=dihedral(4):
> 3&*4;
                                   2
```

Hasil tersebut dapat dibandingkan dengan tabel perkalian (tabel Caley) D_4 atau dengan hasil langsung menggunakan perintah **kalip.**

```
> kalip(3,4,Grup); &-2; &-2&*2; 4&*2; 2&*5&*&-2;
                          2    4    1    1    7
```

Untuk melakukan perhitungan pada grup lain, maka grup lain juga harus diberi nama dengan **Grup**.

```
> Grup:=siklis(12): 3&*4;kalip(3,4,siklis(12)); 2&*5&*&-2;
                              6    6    5
```

Perhatikan, hasil operasi-operasi di atas sangat tergantung pada grup **Grup** yang didefinisikan sebelumnya. Oleh karena itu pemakaian operasi biner &* dan operasi uner &- harus dilakukan secara hati-hati dengan memperhatikan grup yang dipakai. Setiap kali hendak memakai kedua operator tersebut kita harus memberi nama grupnya dengan **Grup**.

Untuk lebih amannya (terutama jika bekerja dengan banyak grup) sebaiknya digunakan fungsi **kalip** dan **invep**, karena pemakaian kedua fungsi tersebut harus menyatakan grupnya secara eksplisit.

Latihan

Pada bagian akhir proses pembelajaran, perlu diberikan latihan bagi mahasiswa untuk menggunakan fungsi-fungsi Maple yang sudah didefinisikan sebelumnya dan untuk mengetahui pemahaman dan penguasaan konsep dan sifat-sifat grup siklis dan dihedral. Contoh latihan berikut ini dapat diberikan kepada mahasiswa untuk dikerjakan sekaligus sebagai alat evaluasi.

1. Gambar elemen-elemen grup D_3 dan C_6.

2. Gambar elemen ke-15 dalam D_{24}.

3. Nyatakan elemen ke-15 dalam D_{24} sebagai sebuah permutasi.

4. Tampilkan tabel Cayley grup D_3 dan C_6. Apakah keduanya Abelian?

5. Hitunglah nilai-nilai $a\,b\,a^{-1}$ untuk semua pasangan elemen *a* dan *b* dari D_3. Hitung pula nilai-nilai tersebut untuk semua pasangan elemen *a* dan *b* dari D_4.

6. Tebak apakah 11&*17&*27&*&-3 dalam D_{15} merupakan sebuah rotasi atau refleksi, dan cek dengan perkalian langsung.

Penyelesaian di bawah ini digunakan untuk mengecek jawaban mahasiswa. Sebaiknya penyelesaian ini diberikan kepada mahasiswa setelah mereka mencoba sendiri menyelesaikan soal-soal di atas.

No. 1 `> Gambar(dihedral(3),1,2,3);Gambar(siklis(6),1,2,3);`

No. 2 `> Gambar(dihedral(24),15,1,1);`

No. 3 `> dihedral(24)[15];`

No. 4 `> D3:=dihedral(3):C6:=siklis(6):d[3]:=cayley(D3);`
`C[6]:=cayley(C6); isAbelian(D3); isAbelian(C6);`

Grup D_3 bukan grup Abelian. Grup C_6 merupakan grup Abelian.

No. 5 `> D3:=dihedral(3):`

```
MD3:=Matrix(nops(D3),(a,b)->kalip(kalip(a,b,D3),invep(a,D3),D3));
D4:=dihedral(4):
MD4:=Matrix(nops(D4),(a,b)->kalip(kalip(a,b,D4),invep(a,D4),D4));
```

No. 6 Elemen ke-3, 11 dan 17 dalam D_{15} merupakan rotasi, sehingga invers 3 juga rotasi, sedangkan elemen ke-27 adalah suatu refleksi. Jadi hasilkali keempatnya kemungkinan merupakan suatu rotasi! Berikut pengecekannya.

```
> Grup:=dihedral(15): 11&*17&*27&*&-3;
> kalip(11,kalip(17,kalip(27,invep(3,Grup),Grup),Grup),Grup);
```

Ternyata benar, hasilkali tersebut adalah elemen ke-4, yang dalam D_{15} adalah suatu rotasi!

Pembahasan

Penjelasan secara manual konsep grup siklis dan grup dihedral memerlukan banyak pekerjaan, karena diperlukan banyak perhitungan, khususnya untuk mengetahui sifat-sifatnya. Selanjutnya, untuk memperjelas konsep kedua grup, diperlukan visualisasi secara geometris, yang jika dilakukan secara manual juga akan membutuhkan banyak waktu dan tenaga. Selain itu, eksplorasi dan perhitungan manual yang dapat dilakukan oleh mahasiswa sangat terbatas untuk grup-grup berukuran kecil dan kemungkinan terjadinya salah hitung cukup besar.

Sebagaimana telah ditunjukkan di atas, dengan menggunakan **Maple,** komputasi dan visualisasi serta eksplorasi sifat-sifat kedua grup tersebut dapat dilakukan secara mudah dan cepat. Setelah diberitahukan perintah-perintah **Maple** yang diperlukan, mahasiswa dapat melakukan eksplorasi sendiri serta menyelesaikan soal-soal atau masalah tentang grup siklis dan grup dihedral dengan menggunakan **Maple**. Ditinjau dari teori belajar, hal ini sesuai dengan prinsip belajar dengan mengerjakan (*learning by doing*) dan akan memperkuat penguasaan konsep (Arsham, 2002).

Maple bukan merupakan software pembelajaran (*courseware*). **Maple** (Waterloo Maple, Inc, 2001a) adalah software komputer untuk matematika yang menyediakan berbagai fasilitas untuk menyelesaikan secara interaktif masalah-masalah aljabar, kalkulus, matematika diskrit, komputasi numerik, dan bidang-bidang lain dalam matematika. **Maple** merupakan paket pemecahan masalah matematika yang lengkap, yang mendukung berbagai operasi matematika seperti analisis numerik, aljabar simbolik – termasuk teori grup, dan grafik. Kemampuan dan kemudahan-kemudahan yang tersedia pada **Maple** menjadikannya sebuah alat yang handal dan fleksibel bagi para penggunanya di bidang pendidikan, riset, dan industri. Tidak sekedar itu, bahkan **Maple** juga merupakan alat belajar dan mengajar yang menakjubkan (Moore, 2001: 9).

Sekalipun **Maple** bukan merupakan software CAI (*Computer-Assisted Instruction*)[1], namun dalam pengertian yang lebih luas **Maple** dapat digunakan sebagai CBE (*Computer-Based Education*) dan CBI (*Computer-Based Instruction*). Menurut Cotton (2001), CBE dan CBI memiliki pengertian yang lebih luas, mencakup semua penggunaan komputer untuk pendidikan, baik dalam bentuk aktivitas dengan komputer secara terpisah maupun aktivitas dengan komputer untuk memperkuat penguasaan materi yang diberikan oleh guru.

Di beberapa negara maju, **Maple** selain digunakan sebagai alat riset di kalangan perguruan tinggi dan industri juga telah digunakan sebagai alat pembelajaran, khususnya untuk bidang matematika, fisika, dan teknik, di sekolah dan perguruan tinggi, (Waterloo Maple, 1998-99:4; Moore, 2001; Waterloo Maple, 2001b; Waterloo Maple, 2002:4). Maple juga telah mendapatkan

[1] CAI merupakah istilah yang lebih sempit dan lebih merujuk kepada penggunaan komputer untuk kegiatan *drill* dan praktek, tutorial, simulasi, atau permainan, baik tanpa atau dengan bimbingan guru. (Cotton, 2001; Doerr, 1979:121; Hopper & Mandell, 1987: 212). Cotton (2001) juga menyebutkan dua istilah lain yang terkait, yakni CMI (*Computer-managed instruction*) dan CEI (*Computer-enriched instruction*). CMI merujuk pada pemakaian komputer untuk mengelola data dan mengambil keputusan instruksional, mengevaluasi hasil belajar siswa dan mengarahkan mereka pada sumber belajar yang sesuai, serta menyimpan catatan kemajuan belajar mereka. CEI adalah aktivitas belajar dengan komputer di mana komputer melakukan (1) menghasilkan data sesuai permintaan siswa untuk melukiskan hubungan di dalam model realitas sosial atau fisik, (2) menjalankan program yang dibuat siswa, atau (3) memberikan pengayaan umum dalam bentuk latihan yang relatif tak terstruktur yang dirancang untuk merangsang dan memotivasi siswa.

penghargaan sebagai software pendidikan, yakni EDDIE (*Educational Software Review Award*), tahun 2000 dari ComputED Learning Lab, sebuah lembaga layanan non-profit di San Diego (Waterloo Maple, 2001b).

Dengan memanfaatkan kemampuan dan fasilitas yang dimiliki **Maple,** model pembelajaran yang dijelaskan di atas dapat digunakan sebagai pelengkap dan penguatan kegiatan perkuliahan kelas rutin. Kegiatan pembelajaran ini dapat diberikan dalam bentuk praktikum yang dapat dilakukan oleh mahasiswa secara mandiri atau dengan bimbingan dosen.

Model pembelajaran ini juga merupakan salah satu bentuk CEI (*Computer-Enriched Instruction*), karena di dalamnya mahasiswa dituntut untuk menuliskan perintah-perintah atau program yang akan dikerjakan oleh Maple untuk melakukan perhitungan, menghasilkan data, maupun menampilkan visualisasi grafis. Dengan demikian mahasiswa akan memiliki pemahaman dan pengalaman belajar secara lebih mendalam, karena menurut Arsham (2001), belajar terjadi jika kita menenggelamkan diri dalam situasi yang memaksa kita untuk melakukan sesuatu.

Model pembelajaran ini dapat disajikan dalam bentuk kegiatan praktikum untuk mata kuliah Teori Grup, Grup Berhingga atau matakuliah lain yang memuat topik grup siklis dan grup dihedral. Melalui model pembelajaran ini, mahasiswa akan memperoleh setidaknya dua hal, yakni: (1) pemahaman konsep grup siklis dan grup dihedral yang lebih mendalam, dan (2) ketrampilan menggunakan komputer, khsusunya software Maple, untuk alat bantu pemecehan masalah matematika.

Pada akhirnya model pembelajaran ini dapat memberikan banyak wawasan baik bagi mahasiswa maupun dosen untuk secara lebih mudah mempelajari konsep-konsep grup siklis dan dihedral. Hal ini disebabkan karena dalam model pembelajaran ini, mahasiswa tidak sekedar belajar di atas kertas, namun dapat langsung melakukan kegiatan komputasi dan visualisasi dengan bantuan komputer.

Di antara keuntungan-keuntungan model pembelajaran ini adalah:

1. tidak memerlukan kemampuan pemrograman yang rumit;
2. perintah-perintah Maple yang digunakan cukup sederhana;
3. baik dosen maupun mahasiswa tidak perlu melakukan pekerjaan manual baik, untuk komputasi maupun visualisasi;
4. visualisasi grup siklis dan dihedral dapat dilakukan dengan mudah sesuai kehendak dan kebutuhan;
5. elemen-elemen suatu grup dapat dinyatakan secara visual, notasi permutasi maupun angka, sesuai kebutuhan;
6. komputasi dapat dilakukan untuk sebarang grup siklis dan dihedral, baik yang berorder kecil maupun besar;
7. pembuatan tabel Cayley dapat dilakukan secara mudah dan sangat cepat;
8. mahasiswa mempunyai kesempatan banyak untuk melakukan eksplorasi dan penyelidikan sifat-sifat suatu grup siklis dan dihedral;

9. opersi perkalian dalam suatu grup dapat dilakukan dengan menggunakan fungsi maupun dengan operator, seperti dalam matematika;
10. penyelidikian sifat-sifat yang lebih lanjut dapat dilakukan dengan menggunakan fungsi dan prosedur Maple yang sudah ada.

Kesimpulan

Topik abstrak seperti konsep grup siklis dan grup dihedral dalam teori grup dapat dijelaskan secara mudah dan visual dengan menggunakan alat bantu program komputer Maple. Sebagaimana telah ditunjukkan dalam model pembelajaran ini, konsep-konsep abstrak tersebut dapat divisualisasikan secara mudah dengan menggunakan Maple. Prosedur-prosedur yang ada pada Maple memungkinkan proses pembelajaran tersebut dilakukan secara mudah. Prosedur dan fungsi-fungsi baru yang diperlukan dapat didefinisikan secara singkat pada Maple untuk melakukan pekerjaan-pekerjaan yang akan memerlukan banyak waktu jika dilakukan secara manual.

Sekalipun Maple adalah software untuk aplikasi matematika, namun dengan kemampuan dan fasilitas yang disediakannya menjadikan Maple dapat digunakan sebagai alat bantu pembelajaran matematika dan khususnya topik abstrak seperti grup siklis dan grup dihedral. Sekalipun tidak dalam bentuk CAI (*Computer-Assisted Instruction*), penggunaan Maple untuk pembelajaran grup siklis dan dihedral merupakan salah satu bentuk CBI (*Computer-Based Instruction*) dan CEI (*Computer-Enriched Instruction*), di mana mahasiswa dituntut lebih aktif menuliskan perintah-perintah yang harus dikerjakan oleh komputer melalui program **Maple**. Dengan demikian, selain akan memperoleh pemahaman mendalam melalui eksplorasi konsep-konsep dan sifat-sifat grup siklis dan dihedral, mahasiswa juga akan mendapatkan ketrampilan komputer, khususnya pemakaian **Maple.** Pada akhirnya mahasiswa akan mendapatkan ketrampilan menggunakan komputer sebagai alat bantu pemecahan masalah matematika.

Beberapa keuntungan model pembelajaran ini, sebagaimana disebutkan di atas, dapat diperoleh mahasiswa maupun dosen. Keuntungan-keuntungan tersebut tidak akan diperoleh melalui pendekatan pembelajaran biasa, yakni kuliah ceramah dan latihan mengerjakan soal secara manual.

Daftar Pustaka

Arsham, Hossein. 2002. *Computer-assisted Learning ,Concepts & Techniques*. http://ubmail.ubalt.edu/~harsham/opre640a/partX.htm

Cotton , Kathleen. 2001. *Computer-Assisted Instruction*. Northwest Regional Educational Laboratory. http://www.nwrel.org/scpd/sirs/5/cu10.html

Doerr, Christine. 1979. *Microcomputer and the 3R's, A Guide for Teachers*. New Jersey, Hayden Book Company, Inc.

Herstein, I.N. 1975. *Topics in Algebra*. second edition. Singapore, John Wiley & Sons.

Mihailovs, Alec. 2001. *Abstract Algebra with Maple*. http://webpages.shepherd.edu/amihailo

Moore, Gregory A. 2001. *Integratting Maple Into the Math Curriculum, A Sensible Guide for Educator*. Waterloo, Waterloo Maple.Inc.

Waterloo Maple Inc. 1999. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Winter 1998-99*

Waterloo Maple Inc. 2001a. *Maple 7 Online Help*. http://www.maplesoft.com/

Waterloo Maple Inc. 2001b. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Spring 2001; Fall 2001;*

Waterloo Maple Inc. 2002. *The Maple Reporter, The Newsletter from Waterloo Maple, Inc. Winter 2002*